Software
'd. bodnar 12-26-09 0 fixed error between 15 and 8 F

'1998 bytes of 2048 before text strings removed

INCLUDE "modedefs.bas"

ansel=0                 'use pins as digital rather than analog

DEFINE OSC 8            'use 8 mhz oscillator

OSCCON = $70            'set clock speed 

option_reg.7=0          'turn on weak pull ups

wpu = %00100000         'weak pull ups on pin GPIO.5 only 

INTCON = %00001100      '00xx11xx to be CCP1 mode

SerialOut       VAR GPIO.0      'pin 7

NotUsed1        VAR GPIO.1      'pin 6

LED             VAR GPIO.2      'pin 5 

NotUsed3        VAR GPIO.3      'pin 4

DQ              VAR
GPIO.4      'pin 3 One-wire Data-Pin "DQ"

ChangeSwitch    VAR GPIO.5      'pin 2

R_Temp          VAR
WORD        'RAW Temperature readings

C_Temp          VAR WORD        'Celsius Temperature

Decimal_C_Temp  VAR WORD        'store just the decimal C temperature

F_Temp          VAR WORD        'Fahrenheit Temperature

G_Temp          VAR WORD        'Generic Temperature for report (either C or F)

NEGPOS          VAR BYTE        'Sign char "-" or "+"

C               VAR BYTE        'Status of I/O port bit

Minimum         VAR BYTE        'minimum temperature (store as C or F)

Maximum         VAR BYTE        'maximum temperature (store as C or F)

Mode            VAR BYTE        'branch variable

Y               VAR BYTE        'for/next variable

Z               VAR BYTE        'temporary variable

Ones            VAR BYTE        'ones

Tens            VAR BYTE        'tens

Hundreds        VAR BYTE        'hundreds

Scale           VAR BIT         '=0 if Fahrenheit / =1 if Celsius

SleepCounter    VAR BYTE        'count times mode shown before sleep

C_neg           VAR BIT         '=1 if cold bit for C set

MinMaxReptFlag  VAR BYTE        '0=temp, 1=min, 2=max- order of minmaxreport

TPause          CON 300

BetweenPause    CON 300

Mode            = 1             'start in Temperature reporting mode

Scale           = 0             '0 =  Fahrenheit

GPIO            = %00110000     'GPIO 4 & 5 as inputs - others as outputs

DATA @0,0,0                     'start with Fahrenheit scale (0) & mode 0

READ 0, Scale                   'read saved scale (F or C)   

READ 1, Mode                    'read mode 

SEROUT SerialOut,n9600,[12,10,13,"Ver 5.0 Bike Blinker +PLUS+",10,13]

SEROUT SerialOut,n9600,["(c) d. bodnar 12-25-09",10,13]

VeryTop:  

LOW LED 

PAUSE 1000

MinMaxReptFlag=0

GOSUB showscale                 'flash "C" or "F" in Morse Code

'Note that minimum & Maximum are stored as temperature + 100 to handle negatives

Minimum         = 255           'store as temp + 100 , below 100 is negative

Maximum         = 0             'gives range of 100 below to 125 above

IF ChangeSwitch = 0 THEN        'if switch held change scale

    Scale = NOT(Scale)

    GOSUB ShowScale

    WRITE 0,Scale               'save current scale to memory

    WHILE ChangeSwitch=0        'stay here till switch released

        GOSUB showscale

    WEND

ENDIF    

start:

IF ChangeSwitch=0 THEN GOTO ChangeMode

BRANCH  Mode, [Temperature, FullBrightFast, FullBrightSlow, FullOn, MinMaxReport]

GOTO start:

Temperature:

GOSUB Start_Convert

'serout serialout,n9600,[10,13,"-+= ",negpos,10,13 ]

IF Scale = 1 THEN 

    G_Temp=C_Temp        '0=fahrenheit, 1=celsius

    ELSE

    G_Temp=F_Temp

ENDIF

Report_Temperature:

IF NEGPOS = "+" THEN

    IF G_Temp + 100 > Maximum THEN    'save new max if new temperature is above old

        Maximum = G_Temp+100

    ENDIF

    ELSE

    IF 100-G_Temp  > Maximum THEN

        Maximum = 100-G_Temp

    ENDIF

ENDIF

IF NEGPOS="+" THEN

    IF G_Temp + 100 < Minimum THEN    'store new minimum temperature if lower

        Minimum = G_Temp+100

    ENDIF

ENDIF

IF NEGPOS="-" THEN

    IF 100-G_Temp< Minimum THEN

        Minimum=100-G_Temp

    ENDIF

ENDIF

SEROUT SerialOut,n9600,["Min/Max = ",#Minimum," ",#Maximum,10,13,10]

HIGH LED:FOR Y=1 TO 20:TOGGLE LED:PAUSE 20:NEXT Y:LOW LED  'fast blink @ start

IF C_neg = 1 AND Scale=1 THEN ' temperature is below zero Ceisius

    NEGPOS = "-"

ENDIF

IF (Scale=0 AND NEGPOS = "-") OR (C_neg=1 AND Scale=1)  THEN 'dim LED after rapid flash to show below zero temperature

    GOSUB minpwm:

ENDIF

IF ChangeSwitch=0 THEN GOTO ChangeMode 'watch for button press

PAUSE 400

'serout serialout,n9600,["G=",#G_Temp," +- ",NEGPOS,10,13]     'sign is for celsius!

Hundreds=G_Temp/100

'serout serialout,n9600,["Gh=",#G_Temp," ",#hundreds,10,13] 

IF Hundreds<>0 THEN ' show hundreds of degrees if temperature is >=100

    FOR Y=1 TO Hundreds:HIGH LED:PAUSE TPause:LOW LED:PAUSE TPause

    IF ChangeSwitch=0 THEN GOTO ChangeMode  'watch for button press

    NEXT Y

    PAUSE BetweenPause:PAUSE BetweenPause

ENDIF

IF ChangeSwitch=0 THEN GOTO ChangeMode     'watch for button press

G_Temp=G_Temp-(Hundreds*100)

Tens=G_Temp/10

'   serout serialout,n9600,["Gt=",#G_Temp," ",#Tens,10,13] 

IF Tens<>0 THEN ' show tens of degrees

    FOR Y=1 TO Tens:HIGH LED:PAUSE TPause:LOW LED:PAUSE TPause

    IF ChangeSwitch=0 THEN GOTO ChangeMode   'watch for button press

    NEXT Y :PAUSE BetweenPause

    PAUSE BetweenPause

ELSE 

    HIGH LED:PAUSE 40:LOW LED:PAUSE 40:PAUSE BetweenPause

    PAUSE BetweenPause

ENDIF

IF ChangeSwitch=0 THEN GOTO ChangeMode  'watch for button press

G_Temp=G_Temp-(Tens*10)

Ones=G_Temp

'serout serialout,n9600,["Go=",#G_Temp," ",#ones,10,13] 

IF Ones<>0 THEN ' show ones of degrees

FOR Y=1 TO Ones:HIGH LED:PAUSE TPause:LOW LED:PAUSE TPause

IF ChangeSwitch=0 THEN GOTO ChangeMode   'watch for button press

NEXT Y:PAUSE BetweenPause

IF ChangeSwitch=0 THEN GOTO ChangeMode  'watch for button press

PAUSE BetweenPause

ELSE 

IF ChangeSwitch=0 THEN GOTO ChangeMode     'watch for button press

HIGH LED:PAUSE 40:LOW LED:PAUSE 40:PAUSE BetweenPause

ENDIF

GOTO start

Start_Convert:

OWOUT DQ, 1, [$CC,$44]

  'Send calculate temperature command

REPEAT                      

    PAUSEUS 25 
  
     
  'Waiting loop until reading complete

    OWIN DQ, 4, [C]


  'Monitor for pin high transition ...   

UNTIL C <> 0 



  ' ... reading complete

OWOUT DQ, 1, [$CC, $BE]

  'Send read scratchpad command

OWIN DQ, 2, [R_Temp.LowByte,R_Temp.HighByte]
'get temperature

IF R_Temp.11 = 1 THEN          'check below zero bit

    C_neg=1

    NEGPOS = "-"

    R_Temp = ~R_Temp + 1

ELSE 

    NEGPOS = "+"

    C_neg=0

ENDIF

C_Temp = R_Temp >> 3        'Get whole # of degrees & first decimal bit

SEROUT SerialOut , n9600,["C =  ",NEGPOS,#C_Temp/2," deg. C",10,13]

F_Temp = C_Temp * 9  

F_Temp = F_Temp / 5

F_Temp=F_Temp/2 
     'remove decimal bit

IF NEGPOS="+" THEN

    F_Temp = F_Temp + 32 

ENDIF

IF NEGPOS ="-" AND C_Temp <= 36 THEN    'was 18 - caused error!

    F_Temp = 32 - F_Temp 

    NEGPOS = "+"

ENDIF

IF NEGPOS="-" AND C_Temp > 36 THEN      'was 18 - caused error!

    F_Temp = F_Temp  - 32

ENDIF

SEROUT SerialOut,n9600,["F =  ",NEGPOS,#F_Temp," deg.F",10,13]

C_Temp = C_Temp/2


RETURN

FullBrightFast: 'full brightness - fast blink

SEROUT SerialOut,n9600,["@f",10,13] 

HIGH LED:PAUSE 60:LOW LED:PAUSE 60:GOTO start:

FullBrightSlow: 'full brightness - slow blink

SEROUT SerialOut,n9600,["@s",10,13] 

HIGH LED:PAUSE 500:LOW LED:PAUSE 500:GOTO start:

FullOn:

SEROUT 0,n9600,[10,13,"@ON",10,13]

HIGH LED

GOTO start:

MinMaxReport:   'shows current, then min, then max

MinMaxReptFlag=MinMaxReptFlag+1

IF MinMaxReptFlag=4 THEN

    MinMaxReptFlag=1

ENDIF

SEROUT SerialOut,n9600,["@MMR=",#MinMaxReptFlag,10,13]

IF MinMaxReptFlag=1 THEN

    GOTO Temperature

ENDIF

IF MinMaxReptFlag=2 THEN

    GOTO MinReport

ENDIF

IF MinMaxReptFlag=3 THEN

    GOTO MaxReport

ENDIF

GOTO MinMaxReport

MinReport:' Min report

IF Minimum-100 >65000 THEN

    Z=100-Minimum

    ELSE

    Z=Minimum-100

ENDIF

SEROUT SerialOut,n9600,["@Mn=",#Z,10,13]  

IF Minimum>=100 THEN

    G_Temp=Minimum - 100

    ELSE 

    G_Temp=100-Minimum

ENDIF

GOSUB MinPWM

GOTO Report_Temperature

MaxReport:   'Max report

IF Maximum-100 >65000 THEN

    Z=100-Maximum

    ELSE

    Z=Maximum-100

ENDIF

SEROUT SerialOut,n9600,["@Mx=",#Z,10,13]

IF Maximum >=100 THEN   'in case starting temp is below zero C or F

    G_Temp=Maximum - 100

    ELSE 

    G_Temp=100-Maximum

ENDIF

IF G_Temp >=100 THEN 

    NEGPOS="+"

ENDIF

GOSUB MaxPWM 

GOTO Report_Temperature

Changemode:

SleepCounter=0

SEROUT SerialOut,n9600,["CMd",10,13]

LOW LED

Mode=Mode+1:IF Mode>4 THEN Mode=0

StayHere:

    SleepCounter=SleepCounter+1    'count number of times reported

    IF SleepCounter >=6 THEN GoToSleep 'if held for 5 repeats goes to sleep

    FOR Y=1 TO Mode+1     'flash mode

        HIGH LED:PAUSE 100:LOW LED:PAUSE 100

    NEXT Y

PAUSE 500

IF ChangeSwitch=0 THEN StayHere:     'stay put till button released

SEROUT SerialOut,n9600,["MD=",#Mode,10,13]

PAUSE 300

WRITE 0,Scale   'save current scale 

WRITE 1,Mode    'save current mode

GOTO start:

ShowScale:

DIT CON 70        'Morse Code DIT

DAH CON DIT*3     'Morse Code DAH

SPACE CON DIT*5   'Morse Code space between Dits & Dahs

''serout serialout,n9600,["Sc=",#scale,10,13]

IF Scale=0 THEN    'Morse Code "F"

    HIGH LED:PAUSE DIT:LOW LED:PAUSE SPACE:HIGH LED:PAUSE DIT:LOW LED:PAUSE SPACE

    HIGH LED:PAUSE DAH:LOW LED:PAUSE SPACE:HIGH LED:PAUSE DIT:LOW LED:PAUSE SPACE

ELSE                'Morse Code "C"

    HIGH LED:PAUSE DAH:LOW LED:PAUSE SPACE:HIGH LED:PAUSE DIT:LOW LED:PAUSE SPACE

    HIGH LED:PAUSE DAH:LOW LED:PAUSE SPACE:HIGH LED:PAUSE DIT:LOW LED:PAUSE SPACE

ENDIF

PAUSE 1000

RETURN

MinPWM:

FOR Y= 255 TO 0 STEP -1

    PWM 2, Y, 3

    IF ChangeSwitch=0 THEN GOTO ChangeMode   'watch for button press

NEXT Y

RETURN

MaxPWM:

FOR Y= 0 TO 255 

    PWM 2, Y, 3                              

    IF ChangeSwitch=0 THEN GOTO ChangeMode  'watch for button press

NEXT Y

RETURN

GoToSleep:

Mode=Mode-1

WRITE 0, Scale      'save current scale 

WRITE 1, Mode       'save current mode

SEROUT 0,n9600,[10,13,"@Sleep",10,13]

FOR Y= 255 TO 0 STEP -1  'dim showing going to sleep

    PWM 2, Y,3

NEXT Y

LOW LED:

INTCON.0 = 0
'Reset the  Port change bit 

IOC=0

LongSleep:

SEROUT 0, n9600, ["Z", 10,13]

PAUSE 3000

INTCON.0 = 0
'Reset the  Port change bit

gpio.5=0        'reset switch pin

IOC = %00100000 'enable INTERRUPT ON CHANGE on gpio 5 

SLEEP 65000

IF intcon.0=0 THEN GOTO LongSleep

INTCON.0 = 0
'Reset the  Port change bit   

IOC=0

FOR Y= 0 TO 255  'brighten LED to show waking up

    PWM 2, Y,3

NEXT Y    

SEROUT 0, n9600, ["xZ", 10,13] 

INTCON.0 = 0
'Reset the  Port change bit   

PAUSE 1000

GOTO VeryTop:

